
ExUnit 
    



      
Unit testing framework for Elixir.

  
    
  
  Example


A basic setup for ExUnit is shown below:
# File: assertion_test.exs

# 1) Start ExUnit.
ExUnit.start()

# 2) Create a new test module (test case) and use "ExUnit.Case".
defmodule AssertionTest do
  # 3) Note that we pass "async: true", this runs the test case
  #    concurrently with other test cases. The individual tests
  #    within each test case are still run serially.
  use ExUnit.Case, async: true

  # 4) Use the "test" macro instead of "def" for clarity.
  test "the truth" do
    assert true
  end
end
To run the tests above, run the file using elixir from the
command line. Assuming you named the file assertion_test.exs,
you can run it as:
$ elixir assertion_test.exs


  
    
  
  Case, Callbacks and Assertions


See ExUnit.Case and ExUnit.Callbacks for more information
about defining test cases and setting up callbacks.
The ExUnit.Assertions module contains a set of macros to
generate assertions with appropriate error messages.

  
    
  
  Integration with Mix


Mix is the project management and build tool for Elixir. Invoking mix test
from the command line will run the tests in each file matching the pattern
*_test.exs found in the test directory of your project.
You must create a test_helper.exs file inside the
test directory and put the code common to all tests there.
The minimum example of a test_helper.exs file would be:
# test/test_helper.exs
ExUnit.start()
Mix will load the test_helper.exs file before executing the tests.
It is not necessary to require the test_helper.exs file in your test
files. Run mix help test for more information.

      


      
        Summary


  
    Types
  


    
      
        failed()

      


        The error state returned by ExUnit.Test and ExUnit.TestModule



    


    
      
        state()

      


        All tests start with a state of nil.



    


    
      
        suite_result()

      


        A map representing the results of running a test suite



    


    
      
        test_id()

      


    





  
    Functions
  


    
      
        after_suite(function)

      


        Sets a callback to be executed after the completion of a test suite.



    


    
      
        async_run()

      


        Starts tests asynchronously while test cases are still loading.



    


    
      
        await_run(task)

      


        Awaits for a test suite that has been started with async_run/0.



    


    
      
        configuration()

      


        Returns ExUnit configuration.



    


    
      
        configure(options)

      


        Configures ExUnit.



    


    
      
        fetch_test_supervisor()

      


        Fetches the test supervisor for the current test.



    


    
      
        plural_rule(word)

      


        Returns the pluralization for word.



    


    
      
        plural_rule(word, pluralization)

      


        Registers a pluralization for word.



    


    
      
        run(additional_modules \\ [])

      


        Runs the tests. It is invoked automatically
if ExUnit is started via start/1.



    


    
      
        start(options \\ [])

      


        Starts ExUnit and automatically runs tests right before the
VM terminates.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    failed()


      
       
       View Source
     


  


  

      

          @type failed() :: [{Exception.kind(), reason :: term(), Exception.stacktrace()}]


      


The error state returned by ExUnit.Test and ExUnit.TestModule

  



  
    
      
      Link to this type
    
    state()


      
       
       View Source
     


  


  

      

          @type state() ::
  nil
  | {:excluded, binary()}
  | {:failed, failed()}
  | {:invalid, module()}
  | {:skipped, binary()}


      


All tests start with a state of nil.
A finished test can be in one of five states:
	Passed (also represented by nil)
	Failed
	Skipped (via @tag :skip)
	Excluded (via :exclude filters)
	Invalid (when setup_all fails)


  



  
    
      
      Link to this type
    
    suite_result()


      
       
       View Source
     


  


  

      

          @type suite_result() :: %{
  excluded: non_neg_integer(),
  failures: non_neg_integer(),
  skipped: non_neg_integer(),
  total: non_neg_integer()
}


      


A map representing the results of running a test suite

  



  
    
      
      Link to this type
    
    test_id()


      
       
       View Source
     


  


  

      

          @type test_id() :: {module(), name :: atom()}


      



  


        

      

      
        Functions

        


  
    
      
      Link to this function
    
    after_suite(function)


      
       
       View Source
     


      (since 1.8.0)

  


  

      

          @spec after_suite((suite_result() -> any())) :: :ok


      


Sets a callback to be executed after the completion of a test suite.
Callbacks set with after_suite/1 must accept a single argument, which is a
map containing the results of the test suite's execution.
If after_suite/1 is called multiple times, the callbacks will be called in
reverse order. In other words, the last callback set will be the first to be
called.

  



  
    
      
      Link to this function
    
    async_run()


      
       
       View Source
     


      (since 1.12.0)

  


  

      

          @spec async_run() :: Task.t()


      


Starts tests asynchronously while test cases are still loading.
It returns a task that must be given to await_run/0 when a result
is desired.

  



  
    
      
      Link to this function
    
    await_run(task)


      
       
       View Source
     


      (since 1.12.0)

  


  

      

          @spec await_run(Task.t()) :: suite_result()


      


Awaits for a test suite that has been started with async_run/0.

  



  
    
      
      Link to this function
    
    configuration()


      
       
       View Source
     


  


  

      

          @spec configuration() :: Keyword.t()


      


Returns ExUnit configuration.
For the available configuration options, see configure/1.

  



  
    
      
      Link to this function
    
    configure(options)


      
       
       View Source
     


  


  

      

          @spec configure(Keyword.t()) :: :ok


      


Configures ExUnit.

  
    
  
  Options


ExUnit supports the following options:
	:assert_receive_timeout - the timeout to be used on assert_receive
calls in milliseconds, defaults to 100;

	:autorun - if ExUnit should run by default on exit. Defaults to true;

	:capture_log - if ExUnit should default to keeping track of log messages
and print them on test failure. Can be overridden for individual tests via
@tag capture_log: false. This can also be configured to a specific level
 with capture_log: [level: LEVEL], for example:
 capture_log: [level: :emergency] to prevent any output from test failures.
 Defaults to false;

	:colors - a keyword list of color options to be used by some formatters:
	:enabled - boolean option to enable colors, defaults to IO.ANSI.enabled?/0;

	:success - success message (defaults to :green)

	:invalid - invalid test message (defaults to :yellow)

	:skipped - skipped test message (defaults to :yellow)

	:failure - failed test message (defaults to :red)

	:error_info - display of actual error (defaults to :red)

	:extra_info - additional information (defaults to :cyan)

	:location_info - filename and tags (defaults to [:bright, :black])

	:diff_insert - color of the insertions on diffs, defaults to :green;

	:diff_insert_whitespace - color of the whitespace insertions on diffs,
defaults to IO.ANSI.color_background(2, 0, 0);

	:diff_delete - color of the deletions on diffs, defaults to :red;

	:diff_delete_whitespace - color of the whitespace deletions on diffs,
defaults to IO.ANSI.color_background(0, 2, 0);



	:exclude - specifies which tests are run by skipping tests that match the
filter. See the "Filters" section in the documentation for ExUnit.Case;

	:exit_status - specifies an alternate exit status to use when the test
suite fails. Defaults to 2;

	:failures_manifest_file - specifies a path to the file used to store failures
between runs;

	:formatters - the formatters that will print results,
defaults to [ExUnit.CLIFormatter];

	:include - specifies which tests are run by skipping tests that do not
match the filter. Keep in mind that all tests are included by default, so unless they are
excluded first, the :include option has no effect. To only run the tests
that match the :include filter, exclude the :test tag first (see the
documentation for ExUnit.Case for more information on tags and filters);

	:max_cases - maximum number of tests to run in parallel. Only tests from
different modules run in parallel. It defaults to System.schedulers_online * 2
to optimize both CPU-bound and IO-bound tests;

	:max_failures - the suite stops evaluating tests when this number of test failures
is reached. All tests within a module that fail when using the
setup_all/1,2 callbacks
are counted as failures. Defaults to :infinity;

	:only_test_ids - a list of {module_name, test_name} tuples that limits
what tests get run. This is typically used by Mix to filter which tests
should run;

	:rand_algorithm - algorithm to be used when generating the test seed.
Available algorithms can be found in Erlang's
:rand documentation (see
:rand.builting_arg/0).
Available since v1.16.0. Before v1.16.0, the algorithm was hard-coded to
:exs1024. On Elixir v1.16.0 and after, the default changed to :exsss;

	:refute_receive_timeout - the timeout to be used on refute_receive
calls in milliseconds, defaults to 100;

	:seed - an integer seed value to randomize the test suite. This seed
is also mixed with the test module and name to create a new unique seed
on every test, which is automatically fed into the :rand module. This
provides randomness between tests, but predictable and reproducible
results. A :seed of 0 will disable randomization and the tests in each
file will always run in the order that they were defined in;

	:slowest - prints timing information for the N slowest tests. Running
ExUnit with slow test reporting automatically runs in trace mode. It
is disabled by default;

	:stacktrace_depth - configures the stacktrace depth to be used
on formatting and reporters, defaults to 20;

	:timeout - sets the timeout for the tests in milliseconds, defaults to 60_000;

	:trace - sets ExUnit into trace mode, this sets :max_cases to 1 and
prints each test case and test while running. Note that in trace mode test
timeouts will be ignored as timeout is set to :infinity;

	:test_location_relative_path - the test location is the file:line information
printed by tests as a shortcut to run a given test. When this value is set,
the value is used as a prefix for the test itself. This is typically used by
Mix to properly set-up umbrella projects;


Any arbitrary configuration can also be passed to configure/1 or start/1,
and these options can then be used in places such as custom formatters. These
other options will be ignored by ExUnit itself.

  



  
    
      
      Link to this function
    
    fetch_test_supervisor()


      
       
       View Source
     


      (since 1.11.0)

  


  

      

          @spec fetch_test_supervisor() :: {:ok, pid()} | :error


      


Fetches the test supervisor for the current test.
Returns {:ok, supervisor_pid} or :error if not called from the test process.
This is the same supervisor as used by ExUnit.Callbacks.start_supervised/2
and similar, see ExUnit.Callbacks module documentation for more information.

  



  
    
      
      Link to this function
    
    plural_rule(word)


      
       
       View Source
     


  


  

      

          @spec plural_rule(binary()) :: binary()


      


Returns the pluralization for word.
If one is not registered, returns the word appended with an "s".

  



  
    
      
      Link to this function
    
    plural_rule(word, pluralization)


      
       
       View Source
     


  


  

      

          @spec plural_rule(binary(), binary()) :: :ok


      


Registers a pluralization for word.
If one is already registered, it is replaced.

  



    

  
    
      
      Link to this function
    
    run(additional_modules \\ [])


      
       
       View Source
     


  


  

      

          @spec run([module()]) :: suite_result()


      


Runs the tests. It is invoked automatically
if ExUnit is started via start/1.
From Elixir v1.14, it accepts an optional list of modules to run
as part of the suite. This is often used to rerun modules already
loaded in memory.
Returns a map containing the total number of tests, the number
of failures, the number of excluded tests and the number of skipped tests.

  



    

  
    
      
      Link to this function
    
    start(options \\ [])


      
       
       View Source
     


  


  

      

          @spec start(Keyword.t()) :: :ok


      


Starts ExUnit and automatically runs tests right before the
VM terminates.
It accepts a set of options to configure ExUnit
(the same ones accepted by configure/1).
If you want to run tests manually, you can set the :autorun option
to false and use run/0 to run tests.

  


        

      



  

    
ExUnit.Assertions 
    



      
This module contains a set of assertion functions that are
imported by default into your test cases.
In general, a developer will want to use the general
assert macro in tests. This macro introspects your code
and provides good reporting whenever there is a failure.
For example, assert some_fun() == 10 will fail (assuming
some_fun() returns 13):
Comparison (using ==) failed in:
code:  assert some_fun() == 10
left:  13
right: 10
This module also provides other convenience functions
like assert_in_delta and assert_raise to easily handle
other common cases such as checking a floating-point number
or handling exceptions.

      


      
        Summary


  
    Functions
  


    
      
        assert(assertion)

      


        Asserts its argument is a truthy value.



    


    
      
        assert(value, message)

      


        Asserts value is truthy, displaying the given message otherwise.



    


    
      
        assert_in_delta(value1, value2, delta, message \\ nil)

      


        Asserts that value1 and value2 differ by no more than delta.



    


    
      
        assert_raise(exception, function)

      


        Asserts the exception is raised during function execution.
Returns the rescued exception, fails otherwise.



    


    
      
        assert_raise(exception, message, function)

      


        Asserts the exception is raised during function execution with
the expected message, which can be a Regex or an exact String.
Returns the rescued exception, fails otherwise.



    


    
      
        assert_receive(pattern, timeout \\ nil, failure_message \\ nil)

      


        Asserts that a message matching pattern was or is going to be received
within the timeout period, specified in milliseconds.



    


    
      
        assert_received(pattern, failure_message \\ nil)

      


        Asserts that a message matching pattern was received and is in the
current process' mailbox.



    


    
      
        catch_error(expression)

      


        Asserts expression will cause an error.



    


    
      
        catch_exit(expression)

      


        Asserts expression will exit.



    


    
      
        catch_throw(expression)

      


        Asserts expression will throw a value.



    


    
      
        flunk(message \\ "Flunked!")

      


        Fails with a message.



    


    
      
        refute(assertion)

      


        A negative assertion, expects the expression to be false or nil.



    


    
      
        refute(value, message)

      


        Asserts value is nil or false (that is, value is not truthy).



    


    
      
        refute_in_delta(value1, value2, delta, message \\ nil)

      


        Asserts value1 and value2 are not within delta.



    


    
      
        refute_receive(pattern, timeout \\ nil, failure_message \\ nil)

      


        Asserts that a message matching pattern was not received (and won't be received)
within the timeout period, specified in milliseconds.



    


    
      
        refute_received(pattern, failure_message \\ nil)

      


        Asserts a message matching pattern was not received (i.e. it is not in the
current process' mailbox).



    





      


      
        Functions

        


  
    
      
      Link to this macro
    
    assert(assertion)


      
       
       View Source
     


      (macro)

  


  

Asserts its argument is a truthy value.
assert introspects the underlying expression and provides
good reporting whenever there is a failure. For example,
if the expression uses the comparison operator, the message
will show the values of the two sides. The assertion
assert 1 + 2 + 3 + 4 > 15
 will fail with the message:
Assertion with > failed
code:  assert 1 + 2 + 3 + 4 > 15
left:  10
right: 15
Similarly, if a match expression is given, it will report
any failure in terms of that match. Given
assert [1] = [2]
you'll see:
match (=) failed
code:  assert [1] = [2]
left: [1]
right: [2]
Keep in mind that assert does not change its semantics
based on the expression. In other words, the expression
is still required to return a truthy value. For example,
the following will fail:
assert nil = some_function_that_returns_nil()
Even though the match works, assert still expects a truth
value. In such cases, simply use ==/2 or match?/2.

  



  
    
      
      Link to this function
    
    assert(value, message)


      
       
       View Source
     


  


  

Asserts value is truthy, displaying the given message otherwise.

  
    
  
  Examples


assert false, "it will never be true"

assert x == :foo, "expected x to be foo"

assert match?({:ok, _}, x), "expected x to match {:ok, _}"

  



    

  
    
      
      Link to this function
    
    assert_in_delta(value1, value2, delta, message \\ nil)


      
       
       View Source
     


  


  

Asserts that value1 and value2 differ by no more than delta.
This difference is inclusive, so the test will pass if the difference
and the delta are equal.

  
    
  
  Examples


assert_in_delta 1.1, 1.5, 0.2
assert_in_delta 10, 15, 2
assert_in_delta 10, 15, 5

  



  
    
      
      Link to this function
    
    assert_raise(exception, function)


      
       
       View Source
     


  


  

Asserts the exception is raised during function execution.
Returns the rescued exception, fails otherwise.

  
    
  
  Examples


assert_raise ArithmeticError, fn ->
  1 + "test"
end

assert_raise RuntimeError, fn ->
  raise "assertion will pass due to this raise"
end

  



  
    
      
      Link to this function
    
    assert_raise(exception, message, function)


      
       
       View Source
     


  


  

Asserts the exception is raised during function execution with
the expected message, which can be a Regex or an exact String.
Returns the rescued exception, fails otherwise.

  
    
  
  Examples


assert_raise ArithmeticError, "bad argument in arithmetic expression", fn ->
  1 + "test"
end

assert_raise RuntimeError, ~r/^today's lucky number is 0\.\d+!$/, fn ->
  raise "today's lucky number is #{:rand.uniform()}!"
end

  



    

    

  
    
      
      Link to this macro
    
    assert_receive(pattern, timeout \\ nil, failure_message \\ nil)


      
       
       View Source
     


      (macro)

  


  

Asserts that a message matching pattern was or is going to be received
within the timeout period, specified in milliseconds.
Unlike assert_received, it has a default timeout
of 100 milliseconds.
The pattern argument must be a match pattern. Flunks with failure_message
if a message matching pattern is not received.

  
    
  
  Examples


assert_receive :hello
Asserts against a larger timeout:
assert_receive :hello, 20_000
You can also match against specific patterns:
assert_receive {:hello, _}

x = 5
assert_receive {:count, ^x}

  



    

  
    
      
      Link to this macro
    
    assert_received(pattern, failure_message \\ nil)


      
       
       View Source
     


      (macro)

  


  

Asserts that a message matching pattern was received and is in the
current process' mailbox.
The pattern argument must be a match pattern. Flunks with failure_message
if a message matching pattern was not received.
Timeout is set to 0, so there is no waiting time.

  
    
  
  Examples


send(self(), :hello)
assert_received :hello

send(self(), :bye)
assert_received :hello, "Oh No!"
** (ExUnit.AssertionError) Oh No!
You can also match against specific patterns:
send(self(), {:hello, "world"})
assert_received {:hello, _}

  



  
    
      
      Link to this macro
    
    catch_error(expression)


      
       
       View Source
     


      (macro)

  


  

Asserts expression will cause an error.
Returns the error or fails otherwise.

  
    
  
  Examples


assert catch_error(error(1)) == 1

  



  
    
      
      Link to this macro
    
    catch_exit(expression)


      
       
       View Source
     


      (macro)

  


  

Asserts expression will exit.
Returns the exit status/message of the current process or fails otherwise.

  
    
  
  Examples


assert catch_exit(exit(1)) == 1
To assert exits from linked processes started from the test, trap exits
with Process.flag/2 and assert the exit message with assert_receive/2.
Process.flag(:trap_exit, true)
pid = spawn_link(fn -> Process.exit(self(), :normal) end)
assert_receive {:EXIT, ^pid, :normal}

  



  
    
      
      Link to this macro
    
    catch_throw(expression)


      
       
       View Source
     


      (macro)

  


  

Asserts expression will throw a value.
Returns the thrown value or fails otherwise.

  
    
  
  Examples


assert catch_throw(throw(1)) == 1

  



    

  
    
      
      Link to this function
    
    flunk(message \\ "Flunked!")


      
       
       View Source
     


  


  

      

          @spec flunk(String.t()) :: no_return()


      


Fails with a message.

  
    
  
  Examples


flunk("This should raise an error")

  



  
    
      
      Link to this macro
    
    refute(assertion)


      
       
       View Source
     


      (macro)

  


  

A negative assertion, expects the expression to be false or nil.
Keep in mind that refute does not change the semantics of
the given expression. In other words, the following will fail:
refute {:ok, _} = some_function_that_returns_error_tuple()
The code above will fail because the = operator always fails
when the sides do not match and refute/2 does not change it.
The correct way to write the refutation above is to use match?/2:
refute match?({:ok, _}, some_function_that_returns_error_tuple())

  
    
  
  Examples


refute age < 0

  



  
    
      
      Link to this function
    
    refute(value, message)


      
       
       View Source
     


  


  

Asserts value is nil or false (that is, value is not truthy).

  
    
  
  Examples


refute true, "This will obviously fail"

  



    

  
    
      
      Link to this function
    
    refute_in_delta(value1, value2, delta, message \\ nil)


      
       
       View Source
     


  


  

Asserts value1 and value2 are not within delta.
This difference is exclusive, so the test will fail if the difference
and the delta are equal.
If you supply message, information about the values will
automatically be appended to it.

  
    
  
  Examples


refute_in_delta 1.1, 1.2, 0.2
refute_in_delta 10, 11, 2

  



    

    

  
    
      
      Link to this macro
    
    refute_receive(pattern, timeout \\ nil, failure_message \\ nil)


      
       
       View Source
     


      (macro)

  


  

Asserts that a message matching pattern was not received (and won't be received)
within the timeout period, specified in milliseconds.
The pattern argument must be a match pattern. Flunks with failure_message
if a message matching pattern is received.

  
    
  
  Examples


refute_receive :bye
Refute received with an explicit timeout:
refute_receive :bye, 1000

  



    

  
    
      
      Link to this macro
    
    refute_received(pattern, failure_message \\ nil)


      
       
       View Source
     


      (macro)

  


  

Asserts a message matching pattern was not received (i.e. it is not in the
current process' mailbox).
The pattern argument must be a match pattern. Flunks with failure_message
if a message matching pattern was received.
Timeout is set to 0, so there is no waiting time.

  
    
  
  Examples


send(self(), :hello)
refute_received :bye

send(self(), :hello)
refute_received :hello, "Oh No!"
** (ExUnit.AssertionError) Oh No!

  


        

      



  

    
ExUnit.Callbacks 
    



      
Defines ExUnit callbacks.
This module defines the setup/1, setup/2, setup_all/1, and
setup_all/2 callbacks, as well as the on_exit/2, start_supervised/2
and stop_supervised/1 functions.
The setup callbacks may be used to define test fixtures
and run any initialization code which help bring the system into a known
state. They are defined via macros and each one can optionally receive a map
with test state and metadata, usually referred to as the context.
Optionally, the context to be used in the tests can be extended by the
setup callbacks by returning a properly structured value (see below).
The setup_all callbacks are invoked only once per module, before any
test is run. All setup callbacks are run before each test. No callback
is run if the test case has no tests or all tests have been filtered out.
setup and setup_all callbacks can be defined by either a block, an atom
naming a local function, a {module, function} tuple, or a list of atoms/tuples.
Both can opt to receive the current context by specifying it
as parameter if defined by a block. Functions used to define a test
setup must accept the context as single argument.
A test module can define multiple setup and setup_all callbacks,
and they are invoked in order of appearance.
start_supervised/2 is used to start processes under a supervisor. The
supervisor is linked to the current test process. The supervisor as well
as all child processes are guaranteed to terminate before any on_exit/2
callback runs.
on_exit/2 callbacks are registered on demand, usually to undo an action
performed by a setup callback. on_exit/2 may also take a reference,
allowing the callback to be overridden in the future. A registered on_exit/2
callback will always run, while failures in setup and setup_all will stop
all remaining setup callbacks from executing.
Finally, setup_all callbacks run in a separate process per module, while
all setup callbacks run in the same process as the test itself. on_exit/2
callbacks always run in a separate process, as implied by their name. The
test process always exits with reason :shutdown, which means any process
linked to the test process will also exit, although asynchronously. Therefore
it is preferred to use start_supervised/2 to guarantee synchronous termination.
Here is a rundown of the life-cycle of the test process:
	the test process is spawned
	it runs setup/2 callbacks
	it runs the test itself
	it stops all supervised processes
	the test process exits with reason :shutdown
	on_exit/2 callbacks are executed in a separate process


  
    
  
  Context


If setup_all or setup return a keyword list, a map, or a tuple in the shape
of {:ok, keyword() | map()}, the keyword list or map will be merged into the
current context and will be available in all subsequent setup_all,
setup, and the test itself.
Returning :ok leaves the context unchanged (in setup and setup_all
callbacks).
Returning anything else from setup_all will force all tests to fail,
while a bad response from setup causes the current test to fail.

  
    
  
  Examples


defmodule AssertionTest do
  use ExUnit.Case, async: true

  # "setup_all" is called once per module before any test runs
  setup_all do
    IO.puts("Starting AssertionTest")

    # Context is not updated here
    :ok
  end

  # "setup" is called before each test
  setup do
    IO.puts("This is a setup callback for #{inspect(self())}")

    on_exit(fn ->
      IO.puts("This is invoked once the test is done. Process: #{inspect(self())}")
    end)

    # Returns extra metadata to be merged into context.
    # Any of the following would also work:
    #
    #     {:ok, %{hello: "world"}}
    #     {:ok, [hello: "world"]}
    #     %{hello: "world"}
    #
    [hello: "world"]
  end

  # Same as above, but receives the context as argument
  setup context do
    IO.puts("Setting up: #{context.test}")

    # We can simply return :ok when we don't want to add any extra metadata
    :ok
  end

  # Setups can also invoke a local or imported function that returns a context
  setup :invoke_local_or_imported_function

  test "always pass" do
    assert true
  end

  test "uses metadata from setup", context do
    assert context[:hello] == "world"
    assert context[:from_named_setup] == true
  end

  defp invoke_local_or_imported_function(context) do
    [from_named_setup: true]
  end
end
It is also common to define your setup as a series of functions,
which are put together by calling setup or setup_all with a
list of function names. Each of these functions receive the context and can
return any of the values allowed in setup blocks:
defmodule ExampleContextTest do
  use ExUnit.Case

  setup [:step1, :step2, :step3, {OtherModule, :step4}]

  defp step1(_context), do: [step_one: true]
  defp step2(_context), do: {:ok, step_two: true} # return values with shape of {:ok, keyword() | map()} allowed
  defp step3(_context), do: :ok # Context not modified

  test "context was modified", context do
    assert context[:step_one] == true
    assert context[:step_two] == true
  end
end
Finally, as discussed in the ExUnit.Case documentation, remember
that the initial context metadata can also be set via @tags, which
can then be accessed in the setup block:
defmodule ExampleTagModificationTest do
  use ExUnit.Case

  setup %{login_as: username} do
    {:ok, current_user: username}
  end

  @tag login_as: "max"
  test "tags modify context", context do
    assert context[:login_as] == "max"
    assert context[:current_user] == "max"
  end
end

      


      
        Summary


  
    Functions
  


    
      
        on_exit(name_or_ref \\ make_ref(), callback)

      


        Registers a callback that runs once the test exits.



    


    
      
        setup(block_or_functions)

      


        Defines a callback to be run before each test in a case.



    


    
      
        setup(context, block)

      


        Defines a callback to be run before each test in a case.



    


    
      
        setup_all(block)

      


        Defines a callback to be run before all tests in a case.



    


    
      
        setup_all(context, block)

      


        Defines a callback to be run before all tests in a case.



    


    
      
        start_link_supervised!(child_spec_or_module, opts \\ [])

      


        Same as start_supervised!/2 but links the started process to the test process.



    


    
      
        start_supervised(child_spec_or_module, opts \\ [])

      


        Starts a child process under the test supervisor.



    


    
      
        start_supervised!(child_spec_or_module, opts \\ [])

      


        Same as start_supervised/2 but returns the PID on success and raises if
not started properly.



    


    
      
        stop_supervised(id)

      


        Stops a child process started via start_supervised/2.



    


    
      
        stop_supervised!(id)

      


        Same as stop_supervised/1 but raises if it cannot be stopped.



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    on_exit(name_or_ref \\ make_ref(), callback)


      
       
       View Source
     


  


  

      

          @spec on_exit(term(), (-> term())) :: :ok


      


Registers a callback that runs once the test exits.
callback is a function that receives no arguments and
runs in a separate process than the caller. Its return
value is irrelevant and is discarded.
on_exit/2 is usually called from setup/1 and setup_all/1
callbacks, often to undo the action performed during the setup.
However, on_exit/2 may also be called dynamically. An "ID" (the
name_or_ref argument) can be used to guarantee that the callback
will be invoked only once. ExUnit uses this term to identify an
on_exit/2 handler: if you want to override a previous handler, for
example, use the same name_or_ref across multiple on_exit/2
calls.
If on_exit/2 is called inside setup/1 or inside a test, it's
executed in a blocking fashion after the test exits and before
running the next test. This means that no other test from the same
test case will be running while the on_exit/2 callback for a
previous test is running. on_exit/2 is executed in a different
process than the test process. On the other hand, if on_exit/2 is
called inside a setup_all/1 callback then callback is executed
after running all tests (see setup_all/1 for more information).

  
    
  
  Examples


setup do
  File.write!("fixture.json", "{}")
  on_exit(fn -> File.rm!("fixture.json") end)
end
You can use the same name_or_ref across multiple on_exit/2 calls
to "override" the registered handler:
setup do
  on_exit(:drop_table, fn ->
    Database.drop_table()
  end)
end

test "a test that shouldn't drop the table" do
  on_exit(:drop_table, fn -> :ok end)
end
Relying too much on overriding callbacks like this can lead to test
cases that are hard to understand and with too many layers of
indirection. However, it can be useful in some cases or for library
authors, for example.

  



  
    
      
      Link to this macro
    
    setup(block_or_functions)


      
       
       View Source
     


      (macro)

  


  

Defines a callback to be run before each test in a case.
Accepts one of these:
	a block
	an atom naming a local function
	a {module, function} tuple
	a list of atoms and {module, function} tuples

Can return values to be merged into the context, to set up the state for
tests. For more details, see the "Context" section shown above.
setup/1 callbacks are executed in the same process as the test process.

  
    
  
  Examples


defp clean_up_tmp_directory(context) do
  # perform setup
  :ok
end

setup :clean_up_tmp_directory

setup [:clean_up_tmp_directory, :another_setup]

setup do
  [conn: Plug.Conn.build_conn()]
end

setup {MyModule, :my_setup_function}

  



  
    
      
      Link to this macro
    
    setup(context, block)


      
       
       View Source
     


      (macro)

  


  

Defines a callback to be run before each test in a case.
This is similar to setup/1, but the first argument is the context.
The block argument can only be a block.
For more details, see the "Context" section shown above.

  
    
  
  Examples


setup context do
  [conn: Plug.Conn.build_conn()]
end

  



  
    
      
      Link to this macro
    
    setup_all(block)


      
       
       View Source
     


      (macro)

  


  

Defines a callback to be run before all tests in a case.
Accepts one of these:
	a block
	an atom naming a local function
	a {module, function} tuple
	a list of atoms and {module, function} tuples

Can return values to be merged into the context, to set up the state for
tests. For more details, see the "Context" section shown above.
setup_all/1 callbacks are executed in a separate process than tests.
All setup_all/1 callbacks are executed in order in the same process.

  
    
  
  On-Exit Handlers


On-exit handlers that you register inside setup_all/1 callbacks
are executed at once after all tests in the module have been run.
They are all executed in the same process, which is a separate
process dedicated to running these handlers. These handlers are
executed in the reverse order of their respective setup_all/1
callbacks.

  
    
  
  Examples


# One-arity function name
setup_all :clean_up_tmp_directory

# A module and function
setup_all {MyModule, :my_setup_function}

# A list of one-arity functions and module/function tuples
setup_all [:clean_up_tmp_directory, {MyModule, :my_setup_function}]

defp clean_up_tmp_directory(_context) do
  # perform setup
  :ok
end

# A block
setup_all do
  [conn: Plug.Conn.build_conn()]
end
The context returned by setup_all/1 will be available in all subsequent
setup_all, setup, and the test itself. For instance, the conn from
the previous example can be accessed as:
test "fetches current users", %{conn: conn} do
  # ...
end

  
    
  
  Handlers


You can define "global" on-exit handlers in setup_all/1 callbacks:
setup_all do
  Database.create_table_for(__MODULE__)

  on_exit(fn ->
    Database.drop_table_for(__MODULE__)
  end)

  :ok
end
The handler in the example above will be executed only once, after
running all tests in the module.

  



  
    
      
      Link to this macro
    
    setup_all(context, block)


      
       
       View Source
     


      (macro)

  


  

Defines a callback to be run before all tests in a case.
Similar as setup_all/1 but also takes a context. The second argument
must be a block. See the "Context" section in the module documentation.

  
    
  
  Examples


setup_all _context do
  [conn: Plug.Conn.build_conn()]
end

  



    

  
    
      
      Link to this function
    
    start_link_supervised!(child_spec_or_module, opts \\ [])


      
       
       View Source
     


      (since 1.14.0)

  


  

      

          @spec start_link_supervised!(
  Supervisor.child_spec() | module() | {module(), term()},
  keyword()
) :: pid()


      


Same as start_supervised!/2 but links the started process to the test process.
If the process that was started crashes, the crash is propagated to the test process,
failing the test and printing the cause of the crash.
Note that if the started process terminates before it is linked to the test process,
this function will exit with reason :noproc.

  



    

  
    
      
      Link to this function
    
    start_supervised(child_spec_or_module, opts \\ [])


      
       
       View Source
     


      (since 1.5.0)

  


  

      

          @spec start_supervised(
  Supervisor.child_spec() | module() | {module(), term()},
  keyword()
) :: Supervisor.on_start_child()


      


Starts a child process under the test supervisor.
It expects a child specification or a module, similar to the ones
given to Supervisor.start_link/2. For example, if your application
starts a supervision tree by running:
Supervisor.start_link([MyServer, {OtherSupervisor, ...}], ...)
You can start those processes under test in isolation by running:
start_supervised(MyServer)
start_supervised({OtherSupervisor, :initial_value})
A keyword list can also be given if there is a need to change
the child specification for the given child process:
start_supervised({MyServer, :initial_value}, restart: :temporary)
See the Supervisor module for a discussion on child specifications
and the available specification keys.
The advantage of starting a process under the test supervisor is that
it is guaranteed to exit before the next test starts. Therefore, you
don't need to remove the process at the end of your tests via
stop_supervised/1. You only need to use stop_supervised/1 if you
want to remove a process from the supervision tree in the middle of a
test, as simply shutting down the process would cause it to be restarted
according to its :restart value.
The started process is not linked to the test process and a crash will
not necessarily fail the test. To start and link a process to guarantee
that any crash would also fail the test use start_link_supervised!/2.
This function returns {:ok, pid} in case of success, otherwise it
returns {:error, reason}.

  



    

  
    
      
      Link to this function
    
    start_supervised!(child_spec_or_module, opts \\ [])


      
       
       View Source
     


      (since 1.6.0)

  


  

      

          @spec start_supervised!(
  Supervisor.child_spec() | module() | {module(), term()},
  keyword()
) :: pid()


      


Same as start_supervised/2 but returns the PID on success and raises if
not started properly.

  



  
    
      
      Link to this function
    
    stop_supervised(id)


      
       
       View Source
     


      (since 1.5.0)

  


  

      

          @spec stop_supervised(id :: term()) :: :ok | {:error, :not_found}


      


Stops a child process started via start_supervised/2.
This function expects the id in the child specification.
For example:
{:ok, _} = start_supervised(MyServer)
:ok = stop_supervised(MyServer)
It returns :ok if there is a supervised process with such
id, {:error, :not_found} otherwise.

  



  
    
      
      Link to this function
    
    stop_supervised!(id)


      
       
       View Source
     


      (since 1.10.0)

  


  

      

          @spec stop_supervised!(id :: term()) :: :ok


      


Same as stop_supervised/1 but raises if it cannot be stopped.

  


        

      



  

    
ExUnit.CaptureIO 
    



      
Functionality to capture IO for testing.

  
    
  
  Examples


defmodule AssertionTest do
  use ExUnit.Case

  import ExUnit.CaptureIO

  test "example" do
    assert capture_io(fn -> IO.puts("a") end) == "a\n"
  end

  test "another example" do
    assert with_io(fn ->
      IO.puts("a")
      IO.puts("b")
      2 + 2
    end) == {4, "a\nb\n"}
  end
end

      


      
        Summary


  
    Functions
  


    
      
        capture_io(fun)

      


        Captures IO generated when evaluating fun.



    


    
      
        capture_io(device_input_or_options, fun)

      


        Captures IO generated when evaluating fun.



    


    
      
        capture_io(device, input_or_options, fun)

      


        Captures IO generated when evaluating fun.



    


    
      
        with_io(fun)

      


        Invokes the given fun and returns the result and captured output.



    


    
      
        with_io(device_input_or_options, fun)

      


        Invokes the given fun and returns the result and captured output.



    


    
      
        with_io(device, input_or_options, fun)

      


        Invokes the given fun and returns the result and captured output.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    capture_io(fun)


      
       
       View Source
     


  


  

      

          @spec capture_io((-> any())) :: String.t()


      


Captures IO generated when evaluating fun.
Returns the binary which is the captured output.
By default, capture_io replaces the group_leader (:stdio)
for the current process. Capturing the group leader is done per
process and therefore can be done concurrently.
However, the capturing of any other named device, such as :stderr,
happens globally and persists until the function has ended. While this means
it is safe to run your tests with async: true in many cases, captured output
may include output from a different test and care must be taken when using
capture_io with a named process asynchronously.
A developer can set a string as an input. The default input is an empty
string. If capturing a named device asynchronously, an input can only be given
to the first capture. Any further capture that is given to a capture on that
device will raise an exception and would indicate that the test should be run
synchronously.
Similarly, once a capture on a named device has begun, the encoding on that
device cannot be changed in a subsequent concurrent capture. An error will
be raised in this case.

  
    
  
  IO devices


You may capture the IO from any registered IO device. The device name given
must be an atom representing the name of a registered process. In addition,
Elixir provides two shortcuts:
	:stdio - a shortcut for :standard_io, which maps to
the current Process.group_leader/0 in Erlang

	:stderr - a shortcut for the named process :standard_error
provided in Erlang



  
    
  
  Options


	:input - An input to the IO device, defaults to "".

	:capture_prompt - Define if prompts (specified as arguments to
IO.get* functions) should be captured. Defaults to true. For
IO devices other than :stdio, the option is ignored.

	:encoding (since v1.10.0) - encoding of the IO device. Allowed
values are :unicode (default) and :latin1.



  
    
  
  Examples


To capture the standard io:
iex> capture_io(fn -> IO.write("john") end) == "john"
true

iex> capture_io("this is input", fn ->
...>   input = IO.gets("> ")
...>   IO.write(input)
...> end) == "> this is input"
true

iex> capture_io([input: "this is input", capture_prompt: false], fn ->
...>   input = IO.gets("> ")
...>   IO.write(input)
...> end) == "this is input"
true
Note it is fine to use == with standard IO, because the content is captured
per test process. However, :stderr is shared across all tests, so you will
want to use =~ instead of == for assertions on :stderr if your tests
are async:
iex> capture_io(:stderr, fn -> IO.write(:stderr, "john") end) =~ "john"
true

iex> capture_io(:standard_error, fn -> IO.write(:stderr, "john") end) =~ "john"
true
In particular, avoid empty captures on :stderr with async tests:
iex> capture_io(:stderr, fn -> :nothing end) == ""
true
Otherwise, if the standard error of any other test is captured, the test will
fail.

  
    
  
  Returning values


As seen in the examples above, capture_io returns the captured output.
If you want to also capture the result of the function executed,
use with_io/2.

  



  
    
      
      Link to this function
    
    capture_io(device_input_or_options, fun)


      
       
       View Source
     


  


  

      

          @spec capture_io(atom() | String.t() | keyword(), (-> any())) :: String.t()


      


Captures IO generated when evaluating fun.
See capture_io/1 for more information.

  



  
    
      
      Link to this function
    
    capture_io(device, input_or_options, fun)


      
       
       View Source
     


  


  

      

          @spec capture_io(atom(), String.t() | keyword(), (-> any())) :: String.t()


      


Captures IO generated when evaluating fun.
See capture_io/1 for more information.

  



  
    
      
      Link to this function
    
    with_io(fun)


      
       
       View Source
     


      (since 1.13.0)

  


  

      

          @spec with_io((-> any())) :: {any(), String.t()}


      


Invokes the given fun and returns the result and captured output.
It accepts the same arguments and options as capture_io/1.

  
    
  
  Examples


{result, output} =
  with_io(fn ->
    IO.puts("a")
    IO.puts("b")
    2 + 2
  end)

assert result == 4
assert output == "a\nb\n"

  



  
    
      
      Link to this function
    
    with_io(device_input_or_options, fun)


      
       
       View Source
     


      (since 1.13.0)

  


  

      

          @spec with_io(atom() | String.t() | keyword(), (-> any())) :: {any(), String.t()}


      


Invokes the given fun and returns the result and captured output.
See with_io/1 for more information.

  



  
    
      
      Link to this function
    
    with_io(device, input_or_options, fun)


      
       
       View Source
     


      (since 1.13.0)

  


  

      

          @spec with_io(atom(), String.t() | keyword(), (-> any())) :: {any(), String.t()}


      


Invokes the given fun and returns the result and captured output.
See with_io/1 for more information.

  


        

      



  

    
ExUnit.CaptureLog 
    



      
Functionality to capture logs for testing.

  
    
  
  Examples


defmodule AssertionTest do
  use ExUnit.Case

  import ExUnit.CaptureLog
  require Logger

  test "example" do
    {result, log} =
      with_log(fn ->
        Logger.error("log msg")
        2 + 2
      end)

    assert result == 4
    assert log =~ "log msg"
  end

  test "check multiple captures concurrently" do
    fun = fn ->
      for msg <- ["hello", "hi"] do
        assert capture_log(fn -> Logger.error(msg) end) =~ msg
      end

      Logger.debug("testing")
    end

    assert capture_log(fun) =~ "hello"
    assert capture_log(fun) =~ "testing"
  end
end

      


      
        Summary


  
    Functions
  


    
      
        capture_log(opts \\ [], fun)

      


        Captures Logger messages generated when evaluating fun.



    


    
      
        with_log(opts \\ [], fun)

      


        Invokes the given fun and returns the result and captured log.



    





      


      
        Functions

        


    

  
    
      
      Link to this function
    
    capture_log(opts \\ [], fun)


      
       
       View Source
     


  


  

      

          @spec capture_log(
  keyword(),
  (-> any())
) :: String.t()


      


Captures Logger messages generated when evaluating fun.
Returns the binary which is the captured output. The captured log
messages will be formatted using Logger.default_formatter/1. Any
option, besides the :level, will be forwarded as an override to
the default formatter.
This function mutes the default logger handler and captures any log
messages sent to Logger from the calling processes. It is possible
to ensure explicit log messages from other processes are captured
by waiting for their exit or monitor signal.
Note that when the async is set to true on use ExUnit.Case,
messages from other tests might be captured. This is OK as long
you consider such cases in your assertions, typically by using
the =~/2 operator to perform partial matches.
It is possible to configure the level to capture with :level,
which will set the capturing level for the duration of the
capture, for instance, if the log level is set to :error, then
any message with the lower level will be ignored.
The default level is nil, which will capture all messages.
Note this setting does not override the overall Logger.level/0 value.
Therefore, if Logger.level/0 is set to a higher level than the one
configured in this function, no message will be captured.
The behaviour is undetermined if async tests change Logger level.
To get the result of the evaluation along with the captured log,
use with_log/2.

  



    

  
    
      
      Link to this function
    
    with_log(opts \\ [], fun)


      
       
       View Source
     


      (since 1.13.0)

  


  

      

          @spec with_log(
  keyword(),
  (-> result)
) :: {result, String.t()}
when result: any()


      


Invokes the given fun and returns the result and captured log.
It accepts the same arguments and options as capture_log/2.

  
    
  
  Examples


{result, log} =
  with_log(fn ->
    Logger.error("log msg")
    2 + 2
  end)

assert result == 4
assert log =~ "log msg"

  


        

      



  

    
ExUnit.Case 
    



      
Helpers for defining test cases.
This module must be used in other modules as a way to configure
and prepare them for testing.
When used, it accepts the following options:
	:async - configures tests in this module to run concurrently with
tests in other modules. Tests in the same module never run concurrently.
It should be enabled only if tests do not change any global state.
Defaults to false.

	:register - when false, does not register this module within
ExUnit server. This means the module won't run when ExUnit suite runs.


use ExUnit.Case
When you use ExUnit.Case, it will import the functionality
from ExUnit.Assertions, ExUnit.Callbacks, ExUnit.DocTest,
and this module itself.


  
    
  
  Examples


defmodule AssertionTest do
  # Use the module
  use ExUnit.Case, async: true

  # The "test" macro is imported by ExUnit.Case
  test "always pass" do
    assert true
  end
end

  
    
  
  Context


All tests receive a context as an argument. The context is particularly
useful for sharing information between callbacks and tests:
defmodule KVTest do
  use ExUnit.Case

  setup do
    {:ok, pid} = KV.start_link()
    {:ok, pid: pid}
  end

  test "stores key-value pairs", context do
    assert KV.put(context[:pid], :hello, :world) == :ok
    assert KV.get(context[:pid], :hello) == :world
  end
end
As the context is a map, it can be pattern matched on to extract
information:
test "stores key-value pairs", %{pid: pid} = _context do
  assert KV.put(pid, :hello, :world) == :ok
  assert KV.get(pid, :hello) == :world
end

  
    
  
  Tags


The context is used to pass information from the callbacks to
the test. In order to pass information from the test to the
callback, ExUnit provides tags.
By tagging a test, the tag value can be accessed in the context,
allowing the developer to customize the test. Let's see an
example:
defmodule FileTest do
  # Changing directory cannot be async
  use ExUnit.Case, async: false

  setup context do
    # Read the :cd tag value
    if cd = context[:cd] do
      prev_cd = File.cwd!()
      File.cd!(cd)
      on_exit(fn -> File.cd!(prev_cd) end)
    end

    :ok
  end

  @tag cd: "fixtures"
  test "reads UTF-8 fixtures" do
    File.read("README.md")
  end
end
In the example above, we have defined a tag called :cd that is
read in the setup callback to configure the working directory the
test is going to run on.
Tags are also very effective when used with case templates
(ExUnit.CaseTemplate) allowing callbacks in the case template
to customize the test behaviour.
Note a tag can be set in two different ways:
@tag key: value
@tag :key       # equivalent to setting @tag key: true
If a tag is given more than once, the last value wins.

  
    
  
  Module and describe tags


A tag can be set for all tests in a module or describe block by
setting @moduletag or @describetag inside each context
respectively:
defmodule ApiTest do
  use ExUnit.Case
  @moduletag :external

  describe "makes calls to the right endpoint" do
    @describetag :endpoint

    # ...
  end
end
If you are setting a @moduletag or @describetag attribute, you must
set them after your call to use ExUnit.Case otherwise you will see
compilation errors.
If the same key is set via @tag, the @tag value has higher
precedence.
The setup_all blocks only receive tags that are set using @moduletag.

  
    
  
  Known tags


The following tags are set automatically by ExUnit and are
therefore reserved:
	:module - the module on which the test was defined

	:file - the file on which the test was defined

	:line - the line on which the test was defined

	:test - the test name

	:async - if the test case is in async mode

	:registered - used for ExUnit.Case.register_attribute/3 values

	:describe - the describe block the test belongs to

	:describe_line - the line the describe block begins on

	:doctest - the module or the file being doctested (if a doctest)

	:doctest_line - the line the doctest was defined (if a doctest)

	:doctest_data - additional metadata about doctests (if a doctest)

	:test_type - the test type used when printing test results.
It is set by ExUnit to :test, :doctest and so on, but is customizable.


The following tags customize how tests behave:
	:capture_log - see the "Log Capture" section below

	:skip - skips the test with the given reason

	:timeout - customizes the test timeout in milliseconds (defaults to 60000).
Accepts :infinity as a timeout value.

	:tmp_dir - (since v1.11.0) see the "Tmp Dir" section below



  
    
  
  Filters


Tags can also be used to identify specific tests, which can then
be included or excluded using filters. The most common functionality
is to exclude some particular tests from running, which can be done
via ExUnit.configure/1:
# Exclude all external tests from running
ExUnit.configure(exclude: [external: true])
From now on, ExUnit will not run any test that has the :external option
set to true. This behaviour can be reversed with the :include option
which is usually passed through the command line:
$ mix test --include external:true

Run mix help test for more information on how to run filters via Mix.
Another use case for tags and filters is to exclude all tests that have
a particular tag by default, regardless of its value, and include only
a certain subset:
ExUnit.configure(exclude: :os, include: [os: :unix])
A given include/exclude filter can be given more than once:
ExUnit.configure(exclude: [os: :unix, os: :windows])
Keep in mind that all tests are included by default, so unless they are
excluded first, the include option has no effect.

  
    
  
  Log Capture


ExUnit can optionally suppress printing of log messages that are generated
during a test. Log messages generated while running a test are captured and
only if the test fails are they printed to aid with debugging.
You can opt into this behaviour for individual tests by tagging them with
:capture_log or enable log capture for all tests in the ExUnit configuration:
ExUnit.start(capture_log: true)
This default can be overridden by @tag capture_log: false or
@moduletag capture_log: false.
Since setup_all blocks don't belong to a specific test, log messages generated
in them (or between tests) are never captured. If you want to suppress these
messages as well, remove the console backend globally by setting:
config :logger, backends: []

  
    
  
  Tmp Dir


ExUnit automatically creates a temporary directory for tests tagged with
:tmp_dir and puts the path to that directory into the test context.
The directory is removed before being created to ensure we start with a blank
slate.
The temporary directory path is unique (includes the test module and test name)
and thus appropriate for running tests concurrently. You can customize the path
further by setting the tag to a string, e.g.: tmp_dir: "my_path", which would
make the final path to be: tmp/<module>/<test>/my_path.
Example:
defmodule MyTest do
  use ExUnit.Case, async: true

  @tag :tmp_dir
  test "with tmp_dir", %{tmp_dir: tmp_dir} do
    assert tmp_dir =~ "with tmp_dir"
    assert File.dir?(tmp_dir)
  end
end
As with other tags, :tmp_dir can also be set as @moduletag and
@describetag.

      


      
        Summary


  
    Types
  


    
      
        env()

      


    





  
    Functions
  


    
      
        describe(message, list)

      


        Describes tests together.



    


    
      
        get_last_registered_test(mod)

      


        Returns the most recently registered test case as an %ExUnit.Test{}
struct.



    


    
      
        register_attribute(env, name, opts \\ [])

      


        Registers a new attribute to be used during ExUnit.Case tests.



    


    
      
        register_describe_attribute(env, name, opts \\ [])

      


        Registers a new describe attribute to be used during ExUnit.Case tests.



    


    
      
        register_module_attribute(env, name, opts \\ [])

      


        Registers a new module attribute to be used during ExUnit.Case tests.



    


    
      
        register_test(map, test_type, name, tags)

          deprecated

      


        Registers a test with the given environment.



    


    
      
        register_test(mod, file, line, test_type, name, tags)

      


        Registers a function to run as part of this case.



    


    
      
        test(message)

      


        Defines a not implemented test with a string.



    


    
      
        test(message, var \\ quote do
  _
end, contents)

      


        Defines a test with message.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    env()


      
       
       View Source
     


  


  

      

          @type env() :: module() | Macro.Env.t()


      



  


        

      

      
        Functions

        


  
    
      
      Link to this macro
    
    describe(message, list)


      
       
       View Source
     


      (since 1.3.0)

      (macro)

  


  

Describes tests together.
Every describe block receives a name which is used as prefix for
upcoming tests. Inside a block, ExUnit.Callbacks.setup/1 may be
invoked and it will define a setup callback to run only for the
current block. The describe name is also added as a tag, allowing
developers to run tests for specific blocks.

  
    
  
  Examples


defmodule StringTest do
  use ExUnit.Case, async: true

  describe "String.downcase/1" do
    test "with ascii characters" do
      assert String.downcase("HELLO") == "hello"
    end

    test "with Unicode" do
      assert String.downcase("HÉLLÒ") == "héllò"
    end
  end
end
When using Mix, you can run all tests in a describe block by name:
$ mix test --only describe:"String.downcase/1"

or by passing the exact line the describe block starts on:
$ mix test path/to/file:123

Note describe blocks cannot be nested. Instead of relying on hierarchy
for composition, developers should build on top of named setups. For
example:
defmodule UserManagementTest do
  use ExUnit.Case, async: true

  describe "when user is logged in and is an admin" do
    setup [:log_user_in, :set_type_to_admin]

    test ...
  end

  describe "when user is logged in and is a manager" do
    setup [:log_user_in, :set_type_to_manager]

    test ...
  end

  defp log_user_in(context) do
    # ...
  end
end
By forbidding hierarchies in favor of named setups, it is straightforward
for the developer to glance at each describe block and know exactly the
setup steps involved.

  



  
    
      
      Link to this function
    
    get_last_registered_test(mod)


      
       
       View Source
     


      (since 1.15.0)

  


  

      

          @spec get_last_registered_test(env()) :: ExUnit.Test.t() | nil


      


Returns the most recently registered test case as an %ExUnit.Test{}
struct.
This is used by third-party utilities to allow compile-time configuration
using test tags without having to explicitly pass the test context at
run-time. It is intended to be invoked in macros before the test module
is compiled.
Raises if called with a module that has already been compiled.

  



    

  
    
      
      Link to this function
    
    register_attribute(env, name, opts \\ [])


      
       
       View Source
     


      (since 1.3.0)

  


  

      

          @spec register_attribute(env(), atom(), keyword()) :: :ok


      


Registers a new attribute to be used during ExUnit.Case tests.
The attribute values will be available through context.registered.
Registered values are cleared after each test/3 similar
to @tag.
This function takes the same options as Module.register_attribute/3.

  
    
  
  Examples


defmodule MyTest do
  use ExUnit.Case

  ExUnit.Case.register_attribute(__MODULE__, :fixtures, accumulate: true)

  @fixtures :user
  @fixtures {:post, insert: false}
  test "using custom attribute", context do
    assert context.registered.fixtures == [{:post, insert: false}, :user]
  end

  test "custom attributes are cleared per test", context do
    assert context.registered.fixtures == []
  end
end

  



    

  
    
      
      Link to this function
    
    register_describe_attribute(env, name, opts \\ [])


      
       
       View Source
     


      (since 1.10.0)

  


  

      

          @spec register_describe_attribute(env(), atom(), keyword()) :: :ok


      


Registers a new describe attribute to be used during ExUnit.Case tests.
The attribute values will be available through context.registered.
Registered values are cleared after each describe/2 similar
to @describetag.
This function takes the same options as Module.register_attribute/3.

  
    
  
  Examples


defmodule MyTest do
  use ExUnit.Case

  ExUnit.Case.register_describe_attribute(__MODULE__, :describe_fixtures, accumulate: true)

  describe "using custom attribute" do
    @describe_fixtures :user
    @describe_fixtures {:post, insert: false}

    test "has attribute", context do
      assert context.registered.describe_fixtures == [{:post, insert: false}, :user]
    end
  end

  describe "custom attributes are cleared per describe" do
    test "doesn't have attributes", context do
      assert context.registered.describe_fixtures == []
    end
  end
end

  



    

  
    
      
      Link to this function
    
    register_module_attribute(env, name, opts \\ [])


      
       
       View Source
     


      (since 1.10.0)

  


  

      

          @spec register_module_attribute(env(), atom(), keyword()) :: :ok


      


Registers a new module attribute to be used during ExUnit.Case tests.
The attribute values will be available through context.registered.
This function takes the same options as Module.register_attribute/3.

  
    
  
  Examples


defmodule MyTest do
  use ExUnit.Case

  ExUnit.Case.register_module_attribute(__MODULE__, :module_fixtures, accumulate: true)

  @module_fixtures :user
  @module_fixtures {:post, insert: false}

  test "using custom attribute", context do
    assert context.registered.module_fixtures == [{:post, insert: false}, :user]
  end

  test "still using custom attribute", context do
    assert context.registered.module_fixtures == [{:post, insert: false}, :user]
  end
end

  



  
    
      
      Link to this function
    
    register_test(map, test_type, name, tags)


      
       
       View Source
     


      (since 1.3.0)

  


    
      This function is deprecated. Use register_test/6 instead.
    


  

Registers a test with the given environment.
This function is deprecated in favor of register_test/6 which performs
better under tight loops by avoiding __ENV__.

  



  
    
      
      Link to this function
    
    register_test(mod, file, line, test_type, name, tags)


      
       
       View Source
     


      (since 1.10.0)

  


  

Registers a function to run as part of this case.
This is used by third-party projects, like QuickCheck, to
implement macros like property/3 that works like test
but instead defines a property. See test/3 implementation
for an example of invoking this function.
The test type will be converted to a string and pluralized for
display. You can use ExUnit.plural_rule/2 to set a custom
pluralization.

  



  
    
      
      Link to this macro
    
    test(message)


      
       
       View Source
     


      (macro)

  


  

Defines a not implemented test with a string.
Provides a convenient macro that allows a test to be defined
with a string, but not yet implemented. The resulting test will
always fail and print a "Not implemented" error message. The
resulting test case is also tagged with :not_implemented.

  
    
  
  Examples


test "this will be a test in future"

  



    

  
    
      
      Link to this macro
    
    test(message, var \\ quote do
  _
end, contents)


      
       
       View Source
     


      (macro)

  


  

Defines a test with message.
The test may also define a pattern, which will be matched
against the test context. For more information on contexts, see
ExUnit.Callbacks.

  
    
  
  Examples


test "true is equal to true" do
  assert true == true
end

  


        

      



  

    
ExUnit.CaseTemplate 
    



      
Defines a module template to be used throughout your test suite.
This is useful when there are a set of setup callbacks or a set
of functions that should be shared between test modules.
Let's imagine that you create a MyCase module that calls use ExUnit.CaseTemplate. When a test case module calls use MyCase, the
following things hold true:
	All the functionality that MyCase would have had available from
use ExUnit.Case is available (same as if MyCase called use ExUnit.Case directly)

	The setup and setup_all callbacks that you define in MyCase
get used in the test case module


The options that you pass to use MyCase get also passed to use ExUnit.Case under the hood. This means you can do things like use MyCase, async: true. You can also access this options in using/2.
use ExUnit.CaseTemplate
When you use ExUnit.CaseTemplate, it will import the functionality
from ExUnit.Assertions, ExUnit.Callbacks, and this module itself.
It will also define a __using__ callback, so the module itself can
be used as a template instead of ExUnit.Case.


  
    
  
  Example


defmodule MyCase do
  use ExUnit.CaseTemplate

  setup do
    IO.puts("This will run before each test that uses this case")
  end
end

defmodule MyTest do
  use MyCase, async: true

  test "truth" do
    assert true
  end
end
If you need to "hook" into use MyCase and do other things as well,
you can use the using/2 macro. See its documentation for more
information and examples.
defmodule MyCase do
  use ExUnit.CaseTemplate

  using do
    quote do
      import MyApp.TestHelpers
    end
  end
end

      


      
        Summary


  
    Functions
  


    
      
        using(var \\ quote do
  _
end, list)

      


        Allows a developer to customize the using block
when the case template is used.



    





      


      
        Functions

        


    

  
    
      
      Link to this macro
    
    using(var \\ quote do
  _
end, list)


      
       
       View Source
     


      (macro)

  


  

Allows a developer to customize the using block
when the case template is used.
You can use an optional var argument when calling using/2. ExUnit
will pass whatever argument you pass to use MyCase as this var argument. See the examples below for clarification.

  
    
  
  Example


defmodule MyCase do
  use ExUnit.CaseTemplate

  using do
    quote do
      # This code is injected into every case that calls "use MyCase"
      alias MyApp.FunModule
    end
  end
end
You can specify an argument to using/2:
defmodule MyCase do
  use ExUnit.CaseTemplate

  using options do
    quote do
      if unquote(options)[:import_helpers] do
        import MyApp.TestHelpers
      end
    end
  end
end
The second argument passed to use MyCase gets forwarded to using/2 too:
defmodule SomeTestCase do
  use MyCase, async: true, import_helpers: true

  test "the truth" do
    # truth/0 comes from MyApp.TestHelpers:
    assert truth()
  end
end
Sharing options with use ExUnit.Case
The second argument that you pass to use MyCase is also passed
as the second argument to use ExUnit.Case.


  


        

      



  

    
ExUnit.DocTest 
    



      
Extract test cases from the documentation.
Doctests allow us to generate tests from code examples found
in @moduledoc and @doc attributes. To do this, invoke the
doctest/1 macro from within your test case and ensure your
code examples are written according to the syntax and guidelines
below.

  
    
  
  Syntax


Every new test starts on a new line, with an iex> prefix.
Multiline expressions can be used by prefixing subsequent lines
with either ...> (recommended) or iex>.
The expected result should start the line after the iex>
and ...> line(s) and be terminated by a newline.

  
    
  
  Examples


To run doctests include them in an ExUnit case with a doctest macro:
defmodule MyModuleTest do
  use ExUnit.Case, async: true
  doctest MyModule
end
The doctest macro loops through all functions and
macros defined in MyModule, parsing their documentation in
search of code examples.
A very basic example is:
iex> 1 + 1
2
Expressions on multiple lines are also supported:
iex> Enum.map([1, 2, 3], fn x ->
...>   x * 2
...> end)
[2, 4, 6]
Multiple results can be checked within the same test:
iex> a = 1
1
iex> a + 1
2
If you want to keep any two tests separate,
add an empty line between them:
iex> a = 1
1

iex> a + 1 # will fail with a `undefined variable "a"` error
2
If you don't want to assert for every result in a doctest, you can omit
the result. You can do so between expressions:
iex> pid = spawn(fn -> :ok end)
iex> is_pid(pid)
true
As well as at the end:
iex> Mod.do_a_call_that_should_not_raise!(...)
This is useful when the result is something variable (like a PID in the
example above) or when the result is a complicated data structure and you
don't want to show it all, but just parts of it or some of its properties.
Similarly to IEx you can use numbers in your "prompts":
iex(1)> [1 + 2,
...(1)>  3]
[3, 3]
This is useful in two cases:
	being able to refer to specific numbered scenarios
	copy-pasting examples from an actual IEx session

You can also select or skip functions when calling
doctest. See the documentation on the :except and :only options below
for more information.

  
    
  
  Opaque types


Some types' internal structures are kept hidden and instead show a
user-friendly structure when inspected. The idiom in
Elixir is to print those data types in the format #Name<...>. Because those
values are treated as comments in Elixir code due to the leading
# sign, they require special care when being used in doctests.
Imagine you have a map that contains a DateTime and is printed as:
%{datetime: #DateTime<2023-06-26 09:30:00+09:00 JST Asia/Tokyo>}
If you try to match on such an expression, doctest will fail to compile.
There are two ways to resolve this.
The first is to rely on the fact that doctest can compare internal
structures as long as they are at the root. So one could write:
iex> map = %{datetime: DateTime.from_naive!(~N[2023-06-26T09:30:00], "Asia/Tokyo")}
iex> map.datetime
#DateTime<2023-06-26 09:30:00+09:00 JST Asia/Tokyo>
Whenever a doctest starts with "#Name<", doctest will perform a string
comparison. For example, the above test will perform the following match:
inspect(map.datetime) == "#DateTime<2023-06-26 09:30:00+09:00 JST Asia/Tokyo>"
Alternatively, since doctest results are actually evaluated, you can have
the DateTime building expression as the doctest result:
iex> %{datetime: DateTime.from_naive!(~N[2023-06-26T09:30:00], "Asia/Tokyo")}
%{datetime: DateTime.from_naive!(~N[2023-06-26T09:30:00], "Asia/Tokyo")}
The downside of this approach is that the doctest result is not really
what users would see in the terminal.

  
    
  
  Exceptions


You can also showcase expressions raising an exception, for example:
iex(1)> raise "some error"
** (RuntimeError) some error
Doctest will look for a line starting with ** ( and it will parse it
accordingly to extract the exception name and message. The exception parser
will consider all following lines part of the exception message until there
is an empty line or there is a new expression prefixed with iex>.
Therefore, it is possible to match on multiline messages as long as there
are no empty lines on the message itself.

  
    
  
  When not to use doctest


In general, doctests are not recommended when your code examples contain
side effects. For example, if a doctest prints to standard output, doctest
will not try to capture the output.
Similarly, doctests do not run in any kind of sandbox. So any module
defined in a code example is going to linger throughout the whole test
suite run.

      


      
        Summary


  
    Functions
  


    
      
        doctest(module, opts \\ [])

      


        Generate test cases from module documentation.



    


    
      
        doctest_file(file, opts \\ [])

      


        Generate test cases from a markdown file.



    





      


      
        Functions

        


    

  
    
      
      Link to this macro
    
    doctest(module, opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Generate test cases from module documentation.
Calling doctest(Module) will generate tests for all doctests found
in the module.

  
    
  
  Options


	:except - generates tests for all functions except those listed
(list of {function, arity} tuples, and/or :moduledoc).

	:only - generates tests only for functions listed
(list of {function, arity} tuples, and/or :moduledoc).

	:import - when true, one can test a function defined in the module
without referring to the module name. However, this is not feasible when
there is a clash with a module like Kernel. In these cases, :import
should be set to false and Module.function(...) should be used instead.

	:tags - a list of tags to apply to all generated doctests.



  
    
  
  Examples


defmodule MyModuleTest do
  use ExUnit.Case
  doctest MyModule, except: [:moduledoc, trick_fun: 1]
end
This macro is auto-imported with every ExUnit.Case.

  



    

  
    
      
      Link to this macro
    
    doctest_file(file, opts \\ [])


      
       
       View Source
     


      (since 1.15.0)

      (macro)

  


  

Generate test cases from a markdown file.

  
    
  
  Options


	:tags - a list of tags to apply to all generated doctests.


  
    
  
  Examples


defmodule ReadmeTest do
  use ExUnit.Case
  doctest_file "README.md"
end
This macro is auto-imported with every ExUnit.Case.

  


        

      



  

    
ExUnit.Filters 
    



      
Conveniences for parsing and evaluating filters.

      


      
        Summary


  
    Types
  


    
      
        ex_unit_opts()

      


    


    
      
        location()

      


    


    
      
        t()

      


    





  
    Functions
  


    
      
        eval(include, exclude, tags, collection)

      


        Evaluates the include and exclude filters against the given tags to
determine if tests should be skipped or excluded.



    


    
      
        failure_info(manifest_file)

      


        Returns a tuple containing useful information about test failures from the
manifest. The tuple contains



    


    
      
        normalize(include, exclude)

      


        Normalizes include and exclude filters to remove duplicates
and keep precedence.



    


    
      
        parse(filters)

      


        Parses the given filters, as one would receive from the command line.



    


    
      
        parse_path(file_path)

      


        Parses filters out of a path.



    


    
      
        parse_paths(file_paths)

      


        Like parse_path/1 but for multiple paths.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    ex_unit_opts()


      
       
       View Source
     


  


  

      

          @type ex_unit_opts() :: [exclude: [:test], include: [location(), ...]] | []


      



  



  
    
      
      Link to this type
    
    location()


      
       
       View Source
     


  


  

      

          @type location() :: {:location, {String.t(), pos_integer() | [pos_integer(), ...]}}


      



  



  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: [{atom(), Regex.t() | String.Chars.t()} | atom()]


      



  


        

      

      
        Functions

        


  
    
      
      Link to this function
    
    eval(include, exclude, tags, collection)


      
       
       View Source
     


  


  

      

          @spec eval(t(), t(), map(), [ExUnit.Test.t()]) ::
  :ok | {:excluded, String.t()} | {:skipped, String.t()}


      


Evaluates the include and exclude filters against the given tags to
determine if tests should be skipped or excluded.
Some filters, like :line, may require the whole test collection to
find the closest line, that's why it must also be passed as an argument.
Filters can either be a regular expression or any data structure
that implements the String.Chars protocol, which is invoked before comparing
the filter with the :tag value.

  
    
  
  Precedence


Tests are first excluded, then included, and then skipped (if any left).
If a :skip tag is found in tags, {:skipped, message} is returned if the test
has been left after the exclude and include filters. Otherwise {:exclude, message}
is returned.
The only exception to this rule is that :skip is found in the include filter,
:ok is returned regardless of whether the test was excluded or not.

  
    
  
  Examples


iex> ExUnit.Filters.eval([foo: "bar"], [:foo], %{foo: "bar"}, [])
:ok

iex> ExUnit.Filters.eval([foo: "bar"], [:foo], %{foo: "baz"}, [])
{:excluded, "due to foo filter"}

  



  
    
      
      Link to this function
    
    failure_info(manifest_file)


      
       
       View Source
     


  


  

      

          @spec failure_info(Path.t()) :: {MapSet.t(Path.t()), MapSet.t(ExUnit.test_id())}


      


Returns a tuple containing useful information about test failures from the
manifest. The tuple contains:
	A set of files that contain tests that failed the last time they ran.
The paths are absolute paths.
	A set of test IDs that failed the last time they ran


  



  
    
      
      Link to this function
    
    normalize(include, exclude)


      
       
       View Source
     


  


  

      

          @spec normalize(t() | nil, t() | nil) :: {t(), t()}


      


Normalizes include and exclude filters to remove duplicates
and keep precedence.

  
    
  
  Examples


iex> ExUnit.Filters.normalize(nil, nil)
{[], []}

iex> ExUnit.Filters.normalize([:foo, :bar, :bar], [:foo, :baz])
{[:foo, :bar], [:baz]}

iex> ExUnit.Filters.normalize([foo: "true"], [:foo])
{[foo: "true"], [:foo]}

iex> ExUnit.Filters.normalize([:foo], [foo: "true"])
{[:foo], []}

iex> ExUnit.Filters.normalize([foo: "true"], [foo: true])
{[foo: "true"], []}

iex> ExUnit.Filters.normalize([foo: true], [foo: "true"])
{[foo: true], []}

iex> ExUnit.Filters.normalize([foo: 1, foo: 1, foo: 2], [])
{[foo: 1, foo: 2], []}

iex> ExUnit.Filters.normalize([], [foo: 1, foo: 1, foo: 2])
{[], [foo: 1, foo: 2]}

  



  
    
      
      Link to this function
    
    parse(filters)


      
       
       View Source
     


  


  

      

          @spec parse([String.t()]) :: t()


      


Parses the given filters, as one would receive from the command line.

  
    
  
  Examples


iex> ExUnit.Filters.parse(["foo:bar", "baz", "line:9", "bool:true"])
[{:foo, "bar"}, :baz, {:line, 9}, {:bool, "true"}]

  



  
    
      
      Link to this function
    
    parse_path(file_path)


      
       
       View Source
     


  


  

      

          @spec parse_path(String.t()) :: {String.t(), ex_unit_opts()}


      


Parses filters out of a path.
Determines whether a given file path (supplied to ExUnit/Mix as arguments
on the command line) includes a line number filter, and if so returns the
appropriate ExUnit configuration options.

  



  
    
      
      Link to this function
    
    parse_paths(file_paths)


      
       
       View Source
     


  


  

      

          @spec parse_paths([String.t()]) :: {[String.t()], ex_unit_opts()}


      


Like parse_path/1 but for multiple paths.
ExUnit filter options are combined.

  


        

      



  

    
ExUnit.Formatter 
    



      
Helper functions for formatting and the formatting protocols.
Formatters are GenServers specified during ExUnit configuration
that receive a series of events as casts.
The following events are possible:
	{:suite_started, opts} -
the suite has started with the specified options to the runner.

	{:suite_finished, times_us} -
the suite has finished. Returns several measurements in microseconds
for running the suite. See t:times_us for more information.

	{:module_started, test_module} -
a test module has started. See ExUnit.TestModule for details.

	{:module_finished, test_module} -
a test module has finished. See ExUnit.TestModule for details.

	{:test_started, test} -
a test has started. See ExUnit.Test for details.

	{:test_finished, test} -
a test has finished. See ExUnit.Test for details.

	{:sigquit, [test | test_module]} -
the VM is going to shutdown. It receives the test cases (or test
module in case of setup_all) still running.


The formatter will also receive the following events but they are deprecated
and should be ignored:
	{:case_started, test_module} -
a test module has started. See ExUnit.TestModule for details.

	{:case_finished, test_module} -
a test module has finished. See ExUnit.TestModule for details.


The full ExUnit configuration is passed as the argument to GenServer.init/1
callback when the formatters are started. If you need to do runtime configuration
of a formatter, you can add any configuration needed by using ExUnit.configure/1
or ExUnit.start/1, and this will then be included in the options passed to
the GenServer.init/1 callback.

      


      
        Summary


  
    Types
  


    
      
        formatter_callback()

      


        A function that this module calls to format various things.



    


    
      
        formatter_callback_diff_key()

      


        Key passed to a formatter callback to format a diff.



    


    
      
        formatter_callback_info_key()

      


        Key passed to a formatter callback to format information.



    


    
      
        id()

      


    


    
      
        test()

      


    


    
      
        times_us()

      


        The times spent on several parts of the test suite.



    


    
      
        width()

      


        Width for formatting.



    





  
    Functions
  


    
      
        format_assertion_diff(assert_error, padding_size, width, formatter)

      


        Formats ExUnit.AssertionError diff.



    


    
      
        format_filters(filters, type)

      


        Formats filters used to constrain cases to be run.



    


    
      
        format_test_all_failure(test_module, failures, counter, width, formatter)

      


        Receives a test module and formats its failure.



    


    
      
        format_test_failure(test, failures, counter, width, formatter)

      


        Receives a test and formats its failures.



    


    
      
        format_times(times)

      


        Formats time taken running the test suite.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    formatter_callback()


      
       
       View Source
     


      (since 1.16.0)

  


  

      

          @type formatter_callback() ::
  (:diff_enabled?, boolean() -> boolean())
  | (formatter_callback_diff_key(), Inspect.Algebra.t() -> Inspect.Algebra.t())
  | (formatter_callback_info_key(), String.t() -> String.t())


      


A function that this module calls to format various things.
You can pass this functions to various functions in this module, and use it
to customize the formatting of the output. For example, ExUnit's CLI formatter
uses this callback to colorize output.

  
    
  
  Keys


The possible keys are:
	:diff_enabled? - whether diffing is enabled. It receives a boolean
indicating whether diffing is enabled by default and returns a boolean
indicating whether diffing should be enabled for the current test.

	:diff_delete and :diff_delete_whitespace - Should format a diff deletion,
with or without whitespace respectively.

	:diff_insert and :diff_insert_whitespace - Should format a diff insertion,
with or without whitespace respectively.

	:extra_info - Should format extra information, such as the "code: " label
that precedes code to show.

	:error_info - Should format error information.

	:test_module_info - Should format test module information. The message returned
when this key is passed precedes messages such as "failure on setup_all callback [...]".

	:test_info - Should format test information.

	:location_info - Should format test location information.

	:stacktrace_info - Should format stacktrace information.

	:blame_diff - Should format a string of code.



  
    
  
  Examples


For example, to format errors as red strings and everything else as is, you could define
a formatter callback function like this:
formatter_callback = fn
  :error_info, msg -> [:red, msg, :reset] |> IO.ANSI.format() |> IO.iodata_to_binary()
  _key, value -> value
end

  



  
    
      
      Link to this type
    
    formatter_callback_diff_key()


      
       
       View Source
     


      (since 1.16.0)

  


  

      

          @type formatter_callback_diff_key() ::
  :diff_delete
  | :diff_delete_whitespace
  | :diff_insert
  | :diff_insert_whitespace


      


Key passed to a formatter callback to format a diff.
See formatter_callback/0.

  



  
    
      
      Link to this type
    
    formatter_callback_info_key()


      
       
       View Source
     


      (since 1.16.0)

  


  

      

          @type formatter_callback_info_key() ::
  :extra_info
  | :error_info
  | :test_module_info
  | :test_info
  | :location_info
  | :stacktrace_info
  | :blame_diff


      


Key passed to a formatter callback to format information.
See formatter_callback/0.

  



  
    
      
      Link to this type
    
    id()


      
       
       View Source
     


  


  

      

          @type id() :: term()


      



  



  
    
      
      Link to this type
    
    test()


      
       
       View Source
     


  


  

      

          @type test() :: ExUnit.Test.t()


      



  



  
    
      
      Link to this type
    
    times_us()


      
       
       View Source
     


  


  

      

          @type times_us() :: %{
  run: pos_integer(),
  async: pos_integer() | nil,
  load: pos_integer() | nil
}


      


The times spent on several parts of the test suite.
The following properties can be computed:
sync = run - (async || 0)
total = run + (load || 0)
async is nil when there are no async tests.
load is nil when the test suite is running and loading
tests concurrently.

  



  
    
      
      Link to this type
    
    width()


      
       
       View Source
     


      (since 1.16.0)

  


  

      

          @type width() :: non_neg_integer() | :infinity


      


Width for formatting.
For example, see format_assertion_diff/4.

  


        

      

      
        Functions

        


  
    
      
      Link to this function
    
    format_assertion_diff(assert_error, padding_size, width, formatter)


      
       
       View Source
     


  


  

      

          @spec format_assertion_diff(
  ExUnit.AssertionError.t(),
  non_neg_integer(),
  width(),
  formatter_callback()
) :: keyword()


      


Formats ExUnit.AssertionError diff.
It returns a keyword list with diffing information
from the left and right side of the assertion, if
any exists.
It expects the assertion error, the padding_size
for formatted content, the width (may be :infinity),
and the formatter callback function.

  
    
  
  Examples


iex> error = assert_raise ExUnit.AssertionError, fn -> assert [1, 2] == [1, 3] end
iex> formatter_cb = fn
...>   :diff_enabled?, _ -> true
...>   _key, value -> value
...> end
iex> keyword = format_assertion_diff(error, 5, 80, formatter_cb)
iex> for {key, val} <- keyword, do: {key, IO.iodata_to_binary(val)}
[left: "[1, 2]", right: "[1, 3]"]

  



  
    
      
      Link to this function
    
    format_filters(filters, type)


      
       
       View Source
     


  


  

      

          @spec format_filters(
  keyword(),
  atom()
) :: String.t()


      


Formats filters used to constrain cases to be run.

  
    
  
  Examples


iex> format_filters([run: true, slow: false], :include)
"Including tags: [run: true, slow: false]"

iex> format_filters([list: [61, 62, 63]], :exclude)
"Excluding tags: [list: [61, 62, 63]]"

  



  
    
      
      Link to this function
    
    format_test_all_failure(test_module, failures, counter, width, formatter)


      
       
       View Source
     


  


  

      

          @spec format_test_all_failure(
  ExUnit.TestModule.t(),
  [failure],
  non_neg_integer(),
  width(),
  formatter_callback()
) :: String.t()
when failure: {atom(), term(), Exception.stacktrace()}


      


Receives a test module and formats its failure.

  
    
  
  Examples


iex> failure = {:error, catch_error(raise "oops"), _stacktrace = []}
iex> formatter_cb = fn _key, value -> value end
iex> test_module = %ExUnit.TestModule{name: Hello}
iex> format_test_all_failure(test_module, [failure], 1, 80, formatter_cb)
"  1) Hello: failure on setup_all callback, all tests have been invalidated\n     ** (RuntimeError) oops\n"

  



  
    
      
      Link to this function
    
    format_test_failure(test, failures, counter, width, formatter)


      
       
       View Source
     


  


  

      

          @spec format_test_failure(
  test(),
  [failure],
  non_neg_integer(),
  width(),
  formatter_callback()
) :: String.t()
when failure: {atom(), term(), Exception.stacktrace()}


      


Receives a test and formats its failures.

  
    
  
  Examples


iex> failure = {:error, catch_error(raise "oops"), _stacktrace = []}
iex> formatter_cb = fn _key, value -> value end
iex> test = %ExUnit.Test{name: :"it works", module: MyTest, tags: %{file: "file.ex", line: 7}}
iex> format_test_failure(test, [failure], 1, 80, formatter_cb)
"  1) it works (MyTest)\n     file.ex:7\n     ** (RuntimeError) oops\n"

  



  
    
      
      Link to this function
    
    format_times(times)


      
       
       View Source
     


  


  

      

          @spec format_times(times_us()) :: String.t()


      


Formats time taken running the test suite.

  
    
  
  Examples


iex> format_times(%{run: 10000, async: nil, load: nil})
"Finished in 0.01 seconds (0.00s async, 0.01s sync)"

iex> format_times(%{run: 10000, async: nil, load: 20000})
"Finished in 0.03 seconds (0.02s on load, 0.00s async, 0.01s sync)"

iex> format_times(%{run: 10000, async: nil, load: 200_000})
"Finished in 0.2 seconds (0.2s on load, 0.00s async, 0.01s sync)"

iex> format_times(%{run: 100_000, async: 50000, load: 200_000})
"Finished in 0.3 seconds (0.2s on load, 0.05s async, 0.05s sync)"

  


        

      



  

    
ExUnit.Test 
    



      
A struct that keeps information about the test.
It is received by formatters and contains the following fields:
	:name - the test name
	:module - the test module
	:state - the finished test state (see ExUnit.state/0)
	:time - the duration in microseconds of the test's runtime
	:tags - the test tags
	:logs - the captured logs


      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: %ExUnit.Test{
  case: module(),
  logs: String.t(),
  module: module(),
  name: atom(),
  state: ExUnit.state(),
  tags: map(),
  time: non_neg_integer()
}


      



  


        

      



  

    
ExUnit.TestModule 
    



      
A struct that keeps information about the test module.
It is received by formatters and contains the following fields:
	:file  - (since v1.11.0) the file of the test module

	:name  - the test module name

	:state - the test error state (see ExUnit.state/0)

	:tests - all tests in this module



      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: %ExUnit.TestModule{
  file: binary(),
  name: module(),
  state: ExUnit.state(),
  tags: map(),
  tests: [ExUnit.Test.t()]
}


      



  


        

      



  

    
ExUnit.AssertionError exception
    



      
Raised to signal an assertion error.
This is used by macros such as ExUnit.Assertions.assert/1.

      


      
        Summary


  
    Types
  


    
      
        t()

      


    





  
    Functions
  


    
      
        no_value()

      


        Indicates no meaningful value for a field.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


      (since 1.16.0)

  


  

      

          @type t() :: %ExUnit.AssertionError{
  __exception__: true,
  args: any(),
  context: any(),
  doctest: any(),
  expr: any(),
  left: any(),
  message: any(),
  right: any()
}


      



  


        

      

      
        Functions

        


  
    
      
      Link to this function
    
    no_value()


      
       
       View Source
     


  


  

      

          @spec no_value() :: atom()


      


Indicates no meaningful value for a field.

  


        

      



  

    
ExUnit.DocTest.Error exception
    



      
Exception raised when there's an error with the syntax or semantics of a doctest.

      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


      (since 1.16.0)

  


  

      

          @type t() :: %ExUnit.DocTest.Error{__exception__: true, message: String.t()}


      



  


        

      



  

    
ExUnit.DuplicateDescribeError exception
    



      
Exception raised to indicate two or more describe blocks with
the same name.

      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


      (since 1.16.0)

  


  

      

          @type t() :: %ExUnit.DuplicateDescribeError{__exception__: true, message: String.t()}


      



  


        

      



  

    
ExUnit.DuplicateTestError exception
    



      
Exception raised to indicate two or more tests with the same name.

      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


      (since 1.16.0)

  


  

      

          @type t() :: %ExUnit.DuplicateTestError{__exception__: true, message: String.t()}


      



  


        

      



  

    
ExUnit.MultiError exception
    



      
Raised to signal multiple errors happened in a test case.

      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


      (since 1.16.0)

  


  

      

          @type t() :: %ExUnit.MultiError{
  __exception__: true,
  errors: [{Exception.kind(), any(), Exception.stacktrace()}]
}


      



  


        

      



  

    
ExUnit.TimeoutError exception
    



      
Exception raised when a test times out.

      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()


      
       
       View Source
     


      (since 1.16.0)

  


  

      

          @type t() :: %ExUnit.TimeoutError{
  __exception__: true,
  timeout: non_neg_integer(),
  type: String.t()
}

